# Example of Laplace Transform and use in ODE

The differential equation:

```
> ODE := diff( y(t), t, t ) = cos(10*t)*exp(t) ;
```

$$ODE := \frac{d^2}{dt^2}\, y(t) = \cos(10\, t)\, e^t \tag{1}$$

Compute the exacty solution

```
> SOL_EXACT := dsolve( { ODE, y(0)=0, D(y)(0)=1 } ) ;
```

$$SOL\_EXACT := y(t) = -\frac{99}{10201}\cos(10\, t)\, e^t + \frac{20}{10201}\sin(10\, t)\, e^t + \frac{100}{101}\, t + \frac{99}{10201} \tag{2}$$

Transform differential equation into an algebraic equation

```
> with(inttrans) : load package with laplace transform inside
> LT := laplace( ODE, t, s ) ;
```

$$LT := s^2\, laplace(y(t), t, s) - D(y)(0) - s\, y(0) = \frac{s-1}{(s-1)^2 + 100} \tag{3}$$

```
> SOL_IN_S := isolate( LT, laplace(y(t), t, s) ) ;
```

$$SOL\_IN\_S := laplace(y(t), t, s) = \frac{\frac{s-1}{(s-1)^2+100} + D(y)(0) + s\, y(0)}{s^2} \tag{4}$$

Set initial conditions

```
> SOL_IN_S_WITH_IC := subs( y(0)=0, D(y)(0)=1, SOL_IN_S ) ;
```

$$SOL\_IN\_S\_WITH\_IC := laplace(y(t), t, s) = \frac{\frac{s-1}{(s-1)^2+100} + 1}{s^2} \tag{5}$$

Find the solution using "inverse" of laplace transform

```
> SOL_BY_LAPLACE := invlaplace( SOL_IN_S_WITH_IC, s, t ) ;
```

$$SOL\_BY\_LAPLACE := y(t) = \frac{99}{10201} + \frac{100}{101}\, t + \frac{1}{10201}\left(-99\cos(10\, t) + 20\sin(10\, t)\right) e^t \tag{6}$$

Check the difference for Laplace and Standard solution of the ODE

```
> simplify( subs( SOL_EXACT, y(t)) - subs( SOL_BY_LAPLACE, y(t)));
```

$$0 \tag{7}$$

# Example of Laplace Transform for solve a Boundary Value Problem

The differential equations

```
> ODE1 := diff( y(t), t) - diff(x(t),t) - y(t) = sin(10*t) ;
  ODE2 := diff( y(t), t) + diff(x(t),t) - x(t) = exp(t) ;
```

$$ODE1 := \frac{d}{dt}\, y(t) - \left(\frac{d}{dt}\, x(t)\right) - y(t) = \sin(10\, t)$$

$$ODE2 := \frac{d}{dt}\, y(t) + \frac{d}{dt}\, x(t) - x(t) = e^{t} \quad \textbf{(8)}$$

The boundary conditions

```
> BC := x(0) = 1, y(1) = 2 ;
```

$$BC := x(0) = 1, y(1) = 2 \quad \textbf{(9)}$$

Compute the exacty solution

```
> SOL_EXACT := simplify(dsolve( { ODE1, ODE2, BC } ),size) ;
```

$$SOL\_EXACT := \Bigg\{ x(t) = \frac{1}{40001}\, \frac{1}{\cos\left(\frac{1}{2}\right)\left(\cosh\left(\frac{1}{2}\right) + \sinh\left(\frac{1}{2}\right)\right)} \Bigg( \Bigg( \Bigg( 38011 \cosh\left(\frac{1}{2}\right) \quad \textbf{(10)}$$

$$+ 38011 \sinh\left(\frac{1}{2}\right)\Bigg) \sin\left(\frac{1}{2}\right) - 2010 \cos(10) - \sin(10) + 40001 \cosh(1)$$

$$+ 40001 \sinh(1) - 80002\Bigg) \sin\left(\frac{1}{2}\, t\right) e^{\frac{1}{2}\, t}\Bigg) + \frac{38011}{40001}\, e^{\frac{1}{2}\, t} \cos\left(\frac{1}{2}\, t\right)$$

$$+ \frac{200}{40001}\, \sin(10\, t) + \frac{1990}{40001}\, \cos(10\, t), y(t)$$

$$= \frac{1}{40001}\, \frac{1}{\cos\left(\frac{1}{2}\right)\left(\cosh\left(\frac{1}{2}\right) + \sinh\left(\frac{1}{2}\right)\right)} \Bigg( \cos\left(\frac{1}{2}\, t\right) e^{\frac{1}{2}\, t} \Bigg( \Bigg( -38011 \cosh\left(\frac{1}{2}\right)$$

$$- 38011 \sinh\left(\frac{1}{2}\right)\Bigg) \sin\left(\frac{1}{2}\right) + 2010 \cos(10) + \sin(10) - 40001 \cosh(1)$$

$$- 40001 \sinh(1) + 80002\Bigg)\Bigg) + \frac{38011}{40001}\, e^{\frac{1}{2}\, t} \sin\left(\frac{1}{2}\, t\right) - \frac{2010}{40001}\, \cos(10\, t)$$

$$- \frac{1}{40001}\, \sin(10\, t) + e^{t} \Bigg\}$$

Transform differential equations into an algebraic system of equations

```
> LT1 := laplace( ODE1, t, s ) ;
  LT2 := laplace( ODE2, t, s ) ;
```

$$LT1 := s\, laplace(y(t), t, s) - y(0) - s\, laplace(x(t), t, s) + x(0) - laplace(y(t), t, s)$$

$$= \frac{10}{s^2 + 100}$$

$$LT2 := s\, laplace(y(t), t, s) - y(0) + s\, laplace(x(t), t, s) - x(0) - laplace(x(t), t, s) \quad \textbf{(11)}$$

$$= \frac{1}{s-1}$$

```
> SOL_IN_S := simplify( solve( {LT1,LT2}, {laplace(x(t), t, s),
  laplace(y(t), t, s)} ), size) ;
```

$$SOL\_IN\_S := \left\{ laplace(x(t), t, s) = \frac{(2 s^3 - s^2 + 200 s - 100) x(0) - s^2 y(0) + s^2 - 100 y(0) - 10 s + 100}{2 s^4 + 201 s^2 - 2 s^3 - 200 s + 100}, laplace(y(t), t, s) = \frac{1}{2} \frac{1}{(s-1)(s^2+100)\left(s^2 - s + \frac{1}{2}\right)} \left( (-300 s + 100 - 3 s^3 + 2 s^4 + 201 s^2) y(0) + (s-1)(s^2+100) x(0) + 10 + 10 s^2 + s^3 + 80 s \right) \right\} \quad (12)$$

Set initial condition (not the final condition y(1)=2)

```
> SOL_IN_S_WITH_IC := subs( BC, SOL_IN_S ) ;
```

$$SOL\_IN\_S\_WITH\_IC := \left\{ laplace(x(t), t, s) = \frac{2 s^3 + 190 s - s^2 y(0) - 100 y(0)}{2 s^4 + 201 s^2 - 2 s^3 - 200 s + 100}, laplace(y(t), t, s) = \frac{1}{2} \frac{1}{(s-1)(s^2+100)\left(s^2 - s + \frac{1}{2}\right)} \left( (-300 s + 100 - 3 s^3 + 2 s^4 + 201 s^2) y(0) + (s-1)(s^2+100) + 10 + 10 s^2 + s^3 + 80 s \right) \right\} \quad (13)$$

Find the solution using "inverse" of laplace transform

```
> SOL_BY_LAPLACE := invlaplace( SOL_IN_S_WITH_IC, s, t ) ;
```

$$SOL\_BY\_LAPLACE := \left\{ x(t) = \frac{1990}{40001} \cos(10 t) + \frac{200}{40001} \sin(10 t) + \frac{1}{40001} \left( 38011 \cos\left(\frac{1}{2} t\right) + (-40001 y(0) + 37991) \sin\left(\frac{1}{2} t\right) \right) e^{\frac{1}{2} t}, y(t) = e^t - \frac{2010}{40001} \cos(10 t) - \frac{1}{40001} \sin(10 t) + \frac{1}{40001} \left( 38011 \sin\left(\frac{1}{2} t\right) + (40001 y(0) - 37991) \cos\left(\frac{1}{2} t\right) \right) e^{\frac{1}{2} t} \right\} \quad (14)$$

```
> SOL_OF_BC := simplify( solve( subs( t=1,subs( SOL_BY_LAPLACE, y(t)
  ))-2, {y(0)} ))  ;
```

$$SOL\_OF\_BC := \left\{ y(0) = \frac{1}{40001} \frac{1}{\cos\left(\frac{1}{2}\right)} \left( \left( -40001\, e + 2010 \cos(10) + \sin(10) - 38011\, e^{\frac{1}{2}} \sin\left(\frac{1}{2}\right) + 37991 \cos\left(\frac{1}{2}\right) e^{\frac{1}{2}} + 80002 \right) e^{-\frac{1}{2}} \right) \right\} \quad (15)$$

```
> SOL_BY_LAPLACE_COMPLETE := subs( SOL_OF_BC, SOL_BY_LAPLACE ) ;
```

$$SOL\_BY\_LAPLACE\_COMPLETE := \left\{ x(t) = \frac{1990}{40001} \cos(10\, t) + \frac{200}{40001} \sin(10\, t) + \frac{1}{40001} \left( 38011 \cos\left(\frac{1}{2}\, t\right) + \left( -\frac{1}{\cos\left(\frac{1}{2}\right)} \left( \left( -40001\, e + 2010 \cos(10) + \sin(10) - 38011\, e^{\frac{1}{2}} \sin\left(\frac{1}{2}\right) + 37991 \cos\left(\frac{1}{2}\right) e^{\frac{1}{2}} + 80002 \right) e^{-\frac{1}{2}} \right) + 37991 \right) \sin\left(\frac{1}{2}\, t\right) \right) e^{\frac{1}{2}\, t}, y(t) = e^{t} - \frac{2010}{40001} \cos(10\, t) - \frac{1}{40001} \sin(10\, t) + \frac{1}{40001} \left( 38011 \sin\left(\frac{1}{2}\, t\right) + \left( \frac{1}{\cos\left(\frac{1}{2}\right)} \left( \left( -40001\, e + 2010 \cos(10) + \sin(10) - 38011\, e^{\frac{1}{2}} \sin\left(\frac{1}{2}\right) + 37991 \cos\left(\frac{1}{2}\right) e^{\frac{1}{2}} + 80002 \right) e^{-\frac{1}{2}} \right) - 37991 \right) \cos\left(\frac{1}{2}\, t\right) \right) e^{\frac{1}{2}\, t} \right\} \quad (16)$$

Check the difference for Laplace and Standard solution of the ODE

```
> simplify( subs( SOL_EXACT, y(t)) - subs( SOL_BY_LAPLACE_COMPLETE, y
  (t)));
```

**(17)**

$$-\frac{1}{40001}\,\frac{1}{\cos\left(\frac{1}{2}\right)\left(\cosh\left(\frac{1}{2}\right)+\sinh\left(\frac{1}{2}\right)\right)}\left(e^{\frac{1}{2}t-\frac{1}{2}}\cos\left(\frac{1}{2}t\right)\left(40001\cosh(1)\,e^{\frac{1}{2}}\right.\right.$$

$$+\,40001\sinh(1)\,e^{\frac{1}{2}}-40001\cosh\left(\frac{1}{2}\right)e-40001\sinh\left(\frac{1}{2}\right)e$$

$$+\,2010\cos(10)\cosh\left(\frac{1}{2}\right)+2010\cos(10)\sinh\left(\frac{1}{2}\right)+\sin(10)\cosh\left(\frac{1}{2}\right)$$

$$+\,\sin(10)\sinh\left(\frac{1}{2}\right)+80002\cosh\left(\frac{1}{2}\right)+80002\sinh\left(\frac{1}{2}\right)-\sin(10)\,e^{\frac{1}{2}}$$

$$\left.\left.-\,2010\cos(10)\,e^{\frac{1}{2}}-80002\,e^{\frac{1}{2}}\right)\right) \quad \textbf{(17)}$$

```
> simplify(convert(%,exp),size);
```

$$0 \quad \textbf{(18)}$$